
9
USING FILES

By defi nition, CL is a control language. Because control languages rarely need to 
process fi les, processing fi les is usually left to more advanced programming lan-
guages. However, CL can process fi les in a limited way, and it can read and write 
to display fi les. Therefore, you can present prompt screens from a CL procedure 
and have user interaction with complete function key support. 

While CL procedures also can read database fi les, writing database fi les is not 
supported. Files can be processed in two different ways. First, they can be pro-
cessed record by record, the same way RPG programs can. Second, they can be 
processed as a whole—like other objects.

RECORD-BY-RECORD PROCESSING OF A FILE

CL procedures can process database and display fi les. Before a fi le can be 
 processed record by record, however, it must be declared to the CL procedure 
 using the Declare File (DCLF) command. A CL procedure can have up to fi ve 
DCLF commands.

The DCLF Command

The DCLF command has three parameters. The fi rst parameter, FILE, identifi es the 
fi le being declared. You must enter the name of the fi le (which can be qualifi ed 
with a library name). The second parameter, RCDFMT, lists the record formats that 
will be known to the CL procedure. It defaults to *ALL to give you a shortcut. If 
you are going to use a fi le (such as a display fi le) that has many record formats, 
but you will use only one or two in your procedure, it is to your advantage to list 
only those record formats you will use in the procedure in the RCDFMT parameter. 
Your program object will become smaller when compiled, and your compile 
listing will be less cluttered. The third parameter, OPNID, is optional, and defaults 
to *NONE, if only one fi le is declared. This value is used to distinguish one fi le 
from another. The OPNID must be referenced in all operations that pertain to the 
fi le, such as RCVF, SNDRCVF, SNDF, ENDRCV, and WAIT.

If your fi le was created from DDS, the CL procedure will know all the fi elds 
contained in the fi le because the CL compiler grabs the external-fi le defi nition. 
The CL procedure can use any of the fi elds as CL variables. For a fi le whose 
OPNID is *NONE (the default value), the variable name is the fi eld name prefi xed 
with an ampersand. For example, if a fi le has a fi eld named TYPE, you can 
expect to fi nd variable &TYPE in your procedure. This defi nition of variables is 
automatic, and you can use &TYPE as you would any other variable. If the OPNID 
parameter has some other value, each variable name consists of the leading 
ampersand, the OPNID value (without trailing blanks), an underscore character, 
and the fi eld name. For example, if a fi le is declared with OPNID(F1) and has two 
fi elds named FILE and LIB, the CL compiler would automatically declare two 
variables named &F1_FILE and &F1_LIB.

If the fi le you are reading was created without DDS, the CL procedure has no 
way of knowing what fi elds you are using in the fi le. Therefore, the compiler 
creates only one variable, that has the same name as the fi le, and declares it 
as type (*CHAR). The variable’s length is the same as the length of the record. 
For  example, suppose you created FILE2 by running the command shown in 
 Figure 9.1.

When you declare the fi le in a CL procedure with the DCLF command, remember 
that you must use variable &FILE2, which will be a 256-byte character string. Use 
the %SST function to extract the parts you need.

The RCVF, WAIT, and ENDRCV Commands

The Receive File (RCVF) command is what CL procedures use to read the fi le 
declared with the DCLF command. RCVF always reads one record. If you process 
several RCVF commands, the fi le will be read sequentially. If the fi le member 
contains no more records, RCVF issues the escape message CPF0864 (end of fi le 
was detected).

Four parameters are defi ned to RCVF, and all of them are optional. The fi rst, DEV, 
allows you to specify the name of a display device from which data is read. By using 
a variable in this parameter, you can retrieve data from different devices by changing 
the value of the variable. The default value for the DEV parameter is *FILE, which 
indicates that data is to be received from the display fi le declared with DCLF.

The second parameter, RCDFMT, is necessary when there are two or more record 
formats in a fi le. You may use the default value, *FILE, when there is only one 
record format in the fi le.

The OPNID parameter specifi es which declared fi le is to be accessed. This value 
must match the OPNID parameter of the DCLF command.

The last parameter, WAIT, applies only to display fi les. WAIT accepts two 
values—*YES (the default) and *NO. WAIT(*YES) causes the system to suspend 
execution of the CL procedure until the input operation is complete. WAIT(*NO) 
causes the system to continue to execute CL commands until it encounters a 
corresponding WAIT command.

CRTPF mylib/file2 RCDLEN(256)

Figure 9.1: An example of creating a fi le with the CRTPF command.

RECORD-BY-RECORD PROCESSING OF A FILE

The WAIT command accepts two, both optional, parameters. The fi rst, DEV, may 
contain a CL variable that contains the name of a display device from which 
input is expected. A value of *NONE means that no device need be specifi ed.

The second parameter, OPNID, must match the OPNID for the DCLF command and 
the corresponding RCVF command in order to link WAIT with the appropriate 
RCVF command.

To cancel a wait state, use the End Receive (ENDRCV) command. Like WAIT, EN-
DRCV has DEV and OPNID parameters. However, the DEV parameter differs in that 
it requires a literal device name rather than a CL variable.

RCVF and Random Input

In combination with the Override with Database File (OVRDBF) command, RCVF 
can read a database fi le randomly by record number, by key, or in reverse sequen-
tial order. All you need to do is use OVRDBF for the fi le being read and indicate 
the desired record in the POSITION parameter. Then use RCVF to read the record. 
For instance, suppose you want to read record number 725 of FILE1. You could 
use the code as shown in Figure 9.2.

Suppose you need to read the record immediately before the record that has a key 
value of ‘ABC’ in a keyed fi le with three key fi elds. You would use code as shown 
in Figure 9.3.

   OVRDBF file1 POSITION(*RRN 725)
   RCVF

Figure 9.2: Reading a specifi c record in a database fi le.

   OVRDBF file1 POSITION(*KEYB 1 FILE1R ‘ABC’)
   RCVF

Figure 9.3: Reading the record prior to key ‘ABC’.

*KEYB indicates that the next RCVF will read a record by key. The “B” means 
that a “key before” type of read is requested. The “1” indicates that you are using 
only one key fi eld in the fi le. FILE1R is the name of the record format that has the 
key. “ABC” is the key value.

RPG programmers can think of OVRDBF as SETGT or SETLL, and RCVF as a READ 
or READP. In the preceding example, OVRDBF acts like SETLL and RCVF acts like 
READP.

The technique just presented (using OVRDBF and RCVF) to emulate random reads 
by key fails in one particular case. If the last key fi eld of the database fi le is of 
type character and is not full (this fi eld’s last byte is blank), RCVF cannot read the 
fi le randomly by key. For example, if the key to the fi le is 4 bytes long and there 
is record with a key “ABC ” (ABC plus a blank), OVRDBF and RCVF won’t be able 
to fi nd this record using *KEY. Use *KEYBE and repeat RCVF until you fi nd the 
record you want.

Note:  Another technique is to use a character variable that is 1 byte 
longer than the key fi eld, and place any non-blank character 
in the last position. Then you can issue an OVRDBF with *KEY 
(Figure 9.4) and it will work okay.

Note:  Variable &KEY is the fi le key (4 bytes long). You can build vari-
able &KEY_PLUS_1 using the key value plus the letter X so the 
last byte is not blank. Then use OVRDBF just like before.

   DCL &key_plus_1 *CHAR    5 /* Key plus one byte: The key length is 4 */

   CHGVAR &key_plus_1 (&key *CAT ‘X’)
   OVRDBF file1 POSITION(*KEY 0 *N &key_plus_1)
   RCVF

Figure 9.4: Positioning a fi le read using a key that is 1 byte longer than the fi le key.

RECORD-BY-RECORD PROCESSING OF A FILE

Notice the use of 0 and *n in the position parameter. These are shortcuts. If  you 
want to use the full key (all key fi elds), use zero instead of the actual number of 
fi elds. And if there is only one record format, you can use *n instead of its name.

The SNDF and SNDRCVF Commands

The Send File (SNDF) command cannot be used on database fi les because CL 
procedures are not allowed to write into them. You can use the SNDF command 
to write a display fi le record, which means an output-only operation to present a 
panel to the interactive user.

If you follow the SNDF command with the RCVF command, you are actually 
displaying a panel and waiting for user input. You also can use the Send Receive 
File (SNDRCVF) command, which combines the two operations in a single com-
mand. In RPG terms, this command would be an EXFMT operation.

If the panel uses any indicators, they become &INXX logical variables to your CL 
procedure. For example, the panel may allow F3=Exit and F12=Cancel. You can 
use the code shown in Figure 9.5 to terminate the procedure.

Another way to code the procedure is shown in Figure 9.6.

   DCLF dspfile

   SNDRCVF RCDFMT(options)

   IF (&in03 *OR &in12) +
      RETURN

Figure 9.5: An example of allowing F3 or F12 to terminate your procedure.

   DCLF dspfile

   SNDRCVF RCDFMT(options)

   IF (&in03 *EQ ‘1’ *OR &in12 *EQ ‘1’) +
      RETURN

Figure 9.6: An alternative method for coding F3 or F12 to terminate your procedure.

The *EQ ‘1’ business is not required because both &IN03 and &IN12 are  logical 
variables. You can use whichever method you like. (You also can test if an 
 indicator is off with the *NOT logical operator. An example would be IF (*NOT 
&IN05) to check whether indicator 05 is off. Using the other coding method, you 
can compare &IN05 against ‘0’.)

Similarly, you can turn indicators on or off before or after you present the panel. 
For example, you can highlight fi elds that have conditioning indicators in the 
DDS of the display fi le for such display attributes as high intensity or underline. 
Remember that any indicators used in the DDS of the display fi le become known 
to the CL procedure as logical variables &IN01 to &IN99. The command shown in 
Figure 9.7 turns on indicator 21.

The CLOSE Command

The CLOSE command closes a fi le that was defi ned with the DCLF command. 
Once the fi le is closed, it is available for processing again. The fi rst RCVF after a 
close of that same fi le re-opens the fi le. CLOSE accepts one parameter—the open 
identifi er. For an example, see Figure 9.8.

   CHGVAR &in21 ‘1’

Figure 9.7: Using CHGVAR command to turn on indicator 21.

RECORD-BY-RECORD PROCESSING OF A FILE

PGM

DCLF   FILE(TempList)  OPNID(Temp)
... omitted lines ...
DOWHILE ‘1’
   RCVF   OPNID(Temp)
   MONMSG   CPF0864  EXEC(LEAVE)   /* end of file */
... omitted lines ...
ENDDO
CLOSE   OPNID(Temp)
/* the file may be re-opened by issuing another RCVF */

Figure 9.8: As of V6R1, you may close a fi le manually.

PROCESSING A FILE AS A WHOLE

CL procedures also can process fi les as another type of object. In this case, the 
entire fi le is processed as a whole and you are not limited to one fi le per CL 
procedure. Furthermore, you don’t use the DCLF command to declare the fi le.

Creating and Deleting Files

CL procedures can be used to create and delete fi les by coding the appropriate 
commands in the CL procedure. For example, you can create a physical fi le with 
the Create Physical File (CRTPF) command and later delete it with the Delete File 
(DLTF) command.

Remember that you don’t need DDS to create a physical fi le if you are willing 
to give up the benefi ts of external fi le defi nition. Keep in mind that creating and 
deleting fi les is a somewhat resource-intensive process that should be kept to a 
minimum. It is more effi cient to create the fi les once and leave them in the libraries.

Processing Database File Members

Database fi les can be processed at the member level with the following 
 commands:

Add Physical File Member (ADDPFM) and Add Logical File Member 
 (ADDLFM).

Change Physical File Member (CHGPFM) and Change Logical File 
 Member (CHGLFM).

Remove Member (RMVM).

Clear Physical File Member (CLRPFM).

Reorganize Physical File Member (RGZPFM).

For example, suppose you have created a work fi le (WORKFILE) for a job. This 
work fi le is not created and deleted each time the procedure runs. To improve 
performance, the work fi le is left in the library between job runs.

Because it exists all the time, the job must ensure that there is no data from 
 previous runs left over. The easiest way to make sure is to run the CLRPFM 
 command to erase all data that exists. If no data exists, CLRPFM will do nothing.

This technique works when the procedure is run by only one person at a time. 
An example would be a program that is always submitted to the same single- 
threaded job queue. Because the job queue is single-threaded, it runs only one job 
at once. But what if the programmer cannot guarantee that two or more people 
won’t attempt to run the same program at the same time?

One solution is to use different members and assign each job a separate member. 
This member must be added (ADDPFM command) at the beginning of the job 
and removed (RMVM command) at the end. Each member must have a different 
name. To assign unique names, you can use the six-digit job number (as shown 
in Figure 9.8).

This method for using temporary fi les is not foolproof; use it with care. The 
program is provided only as an example of using ADDPFM and RMVM. One of 
the problems with this method is that the job might be canceled before it has a 
chance to run RMVM, and that leaves unwanted data in the database fi le.

   DCL &jobnbr     *CHAR    6
   DCL &mbr        *CHAR   10

   RTVJOBA NBR(&jobnbr)
   CHGVAR &mbr (‘JOB’ *CAT &jobnbr)
   ADDPFM FILE(...) MBR(&mbr)

   /* Processing */

   RMVM FILE(...) MBR(&mbr)

Figure 9.8: Assigning a unique name to a member using the six-digit job number.

PROCESSING A FILE AS A WHOLE

The OVRXXXF and DLTOVR Commands

Files can be overridden for a number of reasons. Database fi les, in particular, can 
be overridden for three important reasons:

They are not in any library of the library list. Rather than change the 
library list, the programmer can issue an Override with Database File 
(OVRDBF) command (Figure 9.9), which indicates that fi le FILE1 must be 
obtained from MYLIB.

The HLL program to be called refers to the fi le by a different name. 
Again, the OVRDBF solves this problem. In Figure 9.10, the HLL 
 program uses fi le SOURCE, but the CL programmer wants to process 
 QRPGSRC in library QGPL.

The CL programmer wants the HLL program being called to process a 
particular member of the database fi le—not the fi rst one.

OVRDBF source TOFILE(QGPL/QRGPSRC)
CALL ...

Figure 9.10: Telling a program to use a fi le of another name.

OVRDBF source TOFILE(QGPL/QRGPSRC)
CALL ...

Figure 9.11: Telling a program to use member WORKMBR of database fi le FILE1.

   OVRDBF file1 TOFILE(mylib/file1)

Figure 9.9: Telling the procedure to obtain a fi le that is not in the library list.

In this case, the program being called will process member WORKMBR. Left to 
itself, the system would have processed the fi rst member by default.

Printer and display fi les also can be overridden with the Override with Printer 
File (OVRPRTF) and Override with Display File (OVRDSPF) commands. For 
 example, suppose you created fi le INVPRT to print invoices on form INVOICE, 
which has a form size of 25 lines by 80 columns. Now you need to print invoice 
images on regular stock paper. Run the OVRPRTF command in Figure 9.12 before 
you call the HLL program that prints them.

When you use any of the override commands (OVRXXXF), remember the 
 following rules:

The override applies only to the current and following levels in the call 
stack. If you CALL a program that performs an override, the override will 
no longer be effective when control returns to your program.

The system will activate only one override at each call level. If you 
 override a database fi le with the TOFILE parameter now, and a few 
 statements later perform another override with the MBR parameter, the 
fi le will be overridden only in member. If you suspect the fi le has been 
overridden and want to start over, run the Delete Overrides (DLTOVR) 
command fi rst to delete any overrides currently in effect before you issue 
new ones.

Note:  You can always use the Display Overrides (DSPOVR) command 
to check out what overrides a fi le might have at any moment.

PROCESSING A FILE AS A WHOLE

OVRPRTF invprt FORMTYPE(*STD) PAGESIZE(66 132) OVRFLW(60)

Figure 9.12: Using the OVRPRTF command to change form attributes for the job.

Sorting with OPNQRYF

The Open Query File (OPNQRYF) command probably is the best method to 
sort fi les for later use. With OPNQRYF, you can specify what fi le or fi les to 
use, perform calculations on the fi elds from the fi les used, and select and sort 
the records based on the original fi elds or the calculated results. Because a 
detailed discussion of OPNQRYF fi lls a book by itself, OPNQRYF is described 
briefl y here. For additional information, see MC Press’ Open Query File 
Magic! by Ted Holt.  

OPNQRYF uses the external fi le defi nition; the record layout DDS is provided 
when the fi le is created. You can reference the fi elds by their names (instead 
of by their absolute beginning and ending positions). Doing so gives you 
fl exibility because you don’t have to worry about changing the parameters of 
OPNQRYF if your fi le changes at some future date. The steps to use OPNQRYF 
for one fi le are as follows:

Override the database fi le with SHARE(*YES) using the Override with 
 Database File (OVRDBF) command. This command makes the HLL 
 program that reads the query fi le use a shared open data path (ODP) 
for the fi le in question, rather than creating a new ODP when it opens 
the fi le.

Run OPNQRYF. OPNQRYF creates an open data path over the data fi le(s) 
being queried.

Run your HLL program to make use of the sorted fi le. Your HLL program 
must reference the fi le by its original name and can be coded as keyed or 
sequential access (it makes no difference). Again, because you created 
an open data path, the HLL program will use that data path and not open 
another. The HLL uses the sorted fi le (not the original fi le).

Close the fi le with the Close File (CLOF) command.

Delete the override with the Delete Override (DLTOVR) command.

The following CL program illustrates the entire procedure. An inventory master 
fi le (INVMST) is defi ned in Figure 9.13.

Now suppose you need to write an HLL program—to list the inventory master 
fi le by item number and class code—that includes only the records that have “1” 
in the item type fi eld. You can code the CL program as shown in Figure 9.14.

The OPNQRYF command creates the shared data path. OPNQRYF sorts the fi le by 
item by class (which means that the records will be sorted by item number for 
each item class) and selects only those records that have an item type equal to 
“1”. Your HLL program is executed and the report prints. The CLOF command 
closes the open data path and DLTOVR removes the override. Remember that your 
HLL program must reference the fi le as INVMST.

OVRDBF invmst SHARE(*YES)

   OPNQRYF invmst QRYSLT(‘imtype *EQ ‘’1’’’) +
           KEYFLD((imclas) (imitem))
   CALL ...
   CLOF invmst
   DLTOVR invmst

Figure 9.14: Using OPNQRYF to pre-select the items to be included in a report.

PROCESSING A FILE AS A WHOLE

....+....1....+....2....+....3....+....4....+....5....+....6....+....7
    A         R IMREC
    A           IMITEM     15A       COLHDG(‘Item Number’)
    A           IMTYPE      1A       COLHDG(‘Item’ ‘Type’)
    A           IMCLAS       2A      COLHDG(‘Item’ ‘Class’)
    A           IMCOST       7P  2     COLHDG(‘Standard’ ‘Cost’)
    A                                 EDTCDE(1)
    A           IMPRCE      9P  2     COLHDG(‘Sell’ ‘Price’)
    A                                EDTCDE(1)
    A           IMDESC      30A      COLHDG(‘Item Description’)
    A           IMDRAW      15A       COLHDG(‘Engineering’ ‘Drawing’)
    A           IMLVL       3P  0    COLHDG(‘Level’ ‘Number’)
    A                                EDTCDE(3)
    A           K IMITEM

Figure 9.13: Inventory master fi le (INVMST).

Note:  When OPNQRYF is processing very large fi les and/or selecting 
more than 80 percent of the records in a fi le member, you 
can improve performance by allowing OPNQRYF to use a sort 
routine instead of  access paths. To permit the use of a sort, 
specify ALWCPYDTA(*OPTIMIZE). This does not guarantee that 
OPNQRYF will use a sort routine. The query engine may choose 
to use an  access path instead.

CAPTURING OUTPUT USING QTEMP

QTEMP is a temporary library that exists only between the beginning and the end 
of a job. The system creates a QTEMP library (a different QTEMP library) for each 
job when the job begins, and it deletes the job’s QTEMP library when the job ends. 
Each job has a different QTEMP. If you create an object in your job’s QTEMP, no 
other job in the system can use that object.

QTEMP is the ideal place to put all the objects you create that are needed for the 
duration of the job only. For example, some commands or programs could create 
work fi les that contain data extracted from several fi les, which will be used to 
print a report. The work fi le has no further use after printing the report. Work fi les 
are by no means the only example and *FILE is not the only type of object you 
can put in QTEMP.

Because QTEMP is deleted when the job ends (either normally or abnormally), all 
objects placed in QTEMP will be lost at that time. If you create many objects in 
QTEMP in your interactive session, the SIGNOFF command will take a while to sign 
you off and present the sign-on screen again. During this time, the system will be 
busy deleting the many objects in QTEMP and then QTEMP itself.

Using Permanent Work Files

Frequently, you will need to create work fi les for temporary use. For example, 
you might have to design a report that gathers information from several fi les 
and performs many selections before it prints the report. Resist the temptation 
to create a “fl at” fi le without DDS to describe the fi elds. It might be faster, but 

you lose the capability to use external fi le defi nitions in your procedures. Also, 
program maintenance becomes more diffi cult. Get into the habit of creating 
permanent work fi les in your production libraries. You should put work fi les in 
the same library where you create the programs that use them. For example, you 
can create the fi le MYLIB/WORKFILE after you write the DDS for the fi le.

Each time you need to use the permanent fi le, create a duplicate in QTEMP (as in 
Figure 9.17). The copy in QTEMP will automatically share the complete record 
defi nition you provided with the DDS. In order to use the fi le, all you need to do 
is override it so the QTEMP copy is used instead of the original.

To simplify the task of creating these copies in QTEMP, appendix A includes the 
Create Work File (CRTWRKF) command. Figure 9.18 shows an example of using 
CRTWRKF instead of CRTDUPOBJ.

PGM

   CRTWRKF MODEL(mylib/workfile)

   OVRDBF workfile TOFILE(qtemp/workfile)

   /* Rest of the CL procedure goes here.  All references to WORKFILE */
   /* automatically apply to the copy in QTEMP, not to the original.*/

ENDPGM

Figure 9.18: Example of using the CRTWRKF utility.

CAPTURING OUTPUT USING QTEMP

PGM

   CRTDUPOBJ workfile mylib *file qtemp

   OVRDBF workfile TOFILE(qtemp/workfile)

   /* Rest of the CL procedure goes here.  All references to WORKFILE */
   /* automatically apply to the copy in QTEMP, not to the original.*/

ENDPGM

Figure 9.17: Example of using permanent work fi les.

Because it resides in QTEMP, the system automatically deletes the fi le when the 
job ends.

Using Outfi les

i5/OS commands that begin with the verb Display (DSP) are meant to produce 
output to the display station or to the printer. For example, the Display Program 
(DSPPGM) command shows information about a particular program. Its OUTPUT 
parameter determines where the information is presented. If the user selects * 
(the default value for the OUTPUT parameter), output goes to the display (if the 
command runs interactively) or to the printer (if the command runs in batch). If 
the user selects *PRINT, the output goes to the printer. 

Other commands have a third option named *OUTFILE. In this case, output is 
 directed to a database fi le you specify. For example, consider the command 
shown in Figure 9.19.

The system will run the Display Object Description (DSPOBJD) command but, 
instead of displaying or printing the information gathered, the information goes to 
the fi rst member of fi le QTEMP/OBJECTS and replaces whatever records might have 
been there. If fi le QTEMP/OBJECTS doesn’t exist, it is automatically created.

The command executed (DSPOBJD) contains help text for all parameters. If you 
press the Help key while the cursor is on the OUTFILE parameter, the help text will 
tell you the name of the QSYS fi le that is used as a “model” for the fi le created. 

In the case of the DSPOBJD command, the model outfi le is QSYS/QADSPOBJ. You 
can use the Display File Field Description (DSPFFD) command to get the layout 
of QADSPOBJ’s record. The layout will show you what information you are  getting 
in the outfi le, and where it is within the record. 

   DSPOBJD OBJ(qgpl/*ALL) OBJTYPE(*ALL) OUTPUT(*OUTFILE) +
           OUTFILE(qtemp/objects) OUTMBR(*FIRST *REPLACE)

Figure 9.19: Using an OUTFILE with DSPOBJD.

Outfi les can be processed in CL procedures. For example, suppose you have 
approximately 200 user profi les on your system. For administrative purposes, 
you must change every profi le that belongs to group profi le GRP_A so it belongs 
to group profi le GRP_B. If you performed this task manually, it would be an 
extremely tiresome, time-consuming, and error-prone job. Instead, you can write 
a CL procedure (let’s name it CHGGRPPRF) as shown in Figure 9.20.

Now, execute the program and supply the old and new group profi les. All user 
profi les that had the old user profi le will automatically be changed to the new 
group profi le as shown in Figure 9.21.

CALL chggrpprf (‘GRP_A’ ‘GRP_B’)

Figure 9.21: Calling the program shown in Figure 9.19 and supplying old and new group profi le 
names.

CAPTURING OUTPUT USING QTEMP

PGM (&oldgrpprf &newgrpprf)

   DCL &newgrpprf  *CHAR   10
   DCL &oldgrpprf  *CHAR   10

   DCLF qadspupb

   DSPUSRPRF *ALL TYPE(*BASIC) OUTPUT(*OUTFILE) +
             OUTFILE(qtemp/qadspupb)
   OVRDBF qadspupb TOFILE(qtemp/qadspupb)

loop:
   RCVF
   MONMSG cpf0864 EXEC(RETURN)
   IF (&upgrpf *EQ &oldgrpprf) DO
      CHGUSRPRF USRPRF(&upuprf) GRPPRF(&newgrpprf)
   ENDDO

   GOTO loop

ENDPGM

Figure 9.20: A CL procedure to automatically change the group profi le in every user profi le.

If you encounter this sort of situation often, you might consider creating a com-
mand so that you don’t have to remember what parameters to pass and in what 
order. The program works by running the Display User Profi le (DSPUSRPRF) 
command to an outfi le. Because the outfi le has the same name as the QSYS model 
(QADSPUPB), OVRDBF  runs to make sure that the program uses the fi le in QTEMP 
instead of the empty QSYS model fi le.

The RCVF command is processed and reads a record from QTEMP/QADSPUPB. If 
the end of fi le is detected, CPF0864 is issued (which is trapped by the MONMSG 
command, resulting in an end of the program). Otherwise, the program  compares 
the group profi le name in the user profi le just read (fi eld &UPGRPF from the 
 outfi le) against &OLDGRPPRF. If they are equal, it changes the user profi le 
 (&UPUPRF) to the new group profi le. Variables &UPGRPF and &UPUPRF come from 
the external defi nition of the outfi le.

Capturing OUTPUT(*PRINT)

For those commands that don’t offer the *OUTFILE option in the OUTPUT 
parameter, you can still process the output in a CL procedure if you direct the 
output to *PRINT and then read the printed report in the CL procedure. Here’s 
what you need to do:

Override the printer fi le to HOLD(*YES). You need to know the name of 
the printer fi le that is used by the command whose output you want to 
 process. By holding the fi le, you ensure that it is not printed.

Execute the command that has the output you want to process. Specify 
OUTPUT(*PRINT).

Copy the spooled output to a physical fi le. The physical fi le must already 
exist without an external fi le defi nition. Use the Copy Spooled File 
(CPYSPLF) command to copy the spooled output to the physical fi le.

Now you can process the physical fi le one record at a time. Remember 
that the fi le will contain the report headings, column headings, and footers 

of the normal printed output. These records will have to be recognized and 
ignored by the procedure.

Finally, delete the spooled fi le with the Delete Spooled File (DLTSPLF) 
command and remove the override to the printer fi le with the Delete 
Override (DLTOVR) command.

This process is automated by using the Convert Print to Physical File 
(CVTPRTF) command provided in appendix A.

CAPTURING OUTPUT USING QTEMP




